
Abstract. We derive a Prolog theorem prover for an Intuitionistic Epistemic
Logic by starting from the sequent calculus G4IP that we extend with operator
definitions providing an embedding in intuitionistic propositional logic (IPC).
With help of a candidate definition formula generator, we discover epistemic op-
erators for which axioms and theorems of Artemov and Protopopescu’s Intuition-
istic Epistemic Logic (IEL) hold and formulas expected to be non-theorems fail.
We compare the embedding of IEL in IPC with a similarly discovered successful
embedding of Dosen’s double negation modality, judged inadequate as an epis-
temic operator. Finally, we discuss the failure of the necessitation rule for an
otherwise successful S4 embedding and share our thoughts about the intuitions
explaining these differences between epistemic and alethic modalities in the con-
text of the Brouwer-Heyting-Kolmogorov semantics of intuitionistic reasoning
and knowledge acquisition.

Keywords: epistemic intuitionistic logic, propositional intuitionistic logic, Prolog-based
theorem provers, automatic synthesis of logic systems, definition formula generation al-
gorithms, embedding of modal logics into intuitionistic logic.

1 Introduction

Epistemic Logic Systems have been derived often in parallel and sometime as af-
terthoughts of alethic Modal Logic Systems, in which modalities are defined by axioms
and additional inference rules extending classical logic.

With the advent of Answer Set Programming (ASP) epistemic logics hosted in this
framework like e.g., [1–3] show that intermediate logics 1 can express epistemic oper-
ators by extending the underlying logic with definition of epistemic operators.

Steps2 further below classical logic or ASP are taken in recent work [4], based on
the the Brouwer-Heyting-Kolmogorov (BHK) view of intuitionistic logic that takes into
account the constructive nature of knowledge, modeling more accurately the connection
between proof systems and the related mental processes.

1 Logics stronger than intuitionistic but weaker than classical.
2 Actually infinitely many, as there’s an infinite lattice of intermediate logics between classical

and intuitionistic logic.



Along these lines, our inquiry into epistemic logic will focus on knowledge vs. truth
seen as intuitionistic provability.

Like in the case of embedding epistemic operators into ASP systems, but with a
“machine-learning” twist, we will design a synthesis mechanism for epistemic operators
via embedding in IPC. For this purpose we will generate candidate formulas that verify
axioms, theorems and rules and fail on expected non-theorems. We will also show that
this view generalizes to a mechanism for discovering the right formalization of a given
modal logic.

Our starting point is Artemov and Protopopescu’s Intuitionistic Epistemic Logic
(IEL) [4] that will provide the axioms, theorems and non-theorems stating the require-
ments that must hold for the definitions extending IPC. The discovery mechanism will
also bring up Dosen’s interpretation of double negation [5] as a potential epistemic
operator and we will look into applying the same discovery mechanisms to find an em-
bedding of modal logic S4 in IPC, with special focus on the impact of the necessitation
rule, which requires that all theorems of the logic are necessarily true.

The rest of the paper is organized as follows. Section 2 overviews Artemov and
Protopopescu’s Intuitionistic Epistemic Logic (IEL). Section 3 introduces the G4IP se-
quent calculus prover for Intuitionistic Propositional Logic (IPC). Section 4 describes
the generator for candidate formulas extending IPC with modal operator definitions.
Section 5 explains the discovering of the definitions that ensure the embedding of IEL
into IPC and the discovering of the embedding of Dosen’s double negation as a modal-
ity operator. It also discusses the intuitions behind the embedding of IEL, including the
epistemic equivalent of the necessity rule in IPC and the adequacy of this embedding
as a constructive mechanism for reasoning about knowledge. Section 6 studies the case
of the S4 modal logic and the failure of the necessity rule, indicating the difficulty of
embedding it in IPC by contrast to IEL. Section 7 overviews some related work and
section 8 concludes the paper.

The paper is written as a literate SWI-Prolog program with its extracted code at
https://raw.githubusercontent.com/ptarau/TypesAndProofs/master/ieltp.pro.

2 Overview of Artemov and Protopopescu’s IEL logic

In [4] a system for Intuitionistic Epistemic Logic is introduced that

“maintains the original Brouwer-Heyting-Kolmogorov semantics for intuition-
ism and is consistent with the well-known approach that intuitionistic knowl-
edge be regarded as the result of verification”.

Instead of the classic, alethic-modalities inspired K operator for which

KA→ A

Artemov and Protopopescu argue that co-reflection expresses better the idea of con-
structivity of truth

A→KA

They also argue that this applies to both belief and knowledge i.e., that



“The verification-based approach allows that justifications more general than
proof can be adequate for belief and knowledge”.

On the other hand, they consider intuitionistic reflection acceptable, expressing the
fact that “known propositions cannot be false”:

KA→¬¬A

Thus, they position intuitionistic knowledge of A between A and ¬¬A:

A→KA→¬¬A

and given that (via Glivenko’s transformation [6]) applying double negation to a for-
mula embeds classical propositional calculus into IPC, they express this view as:

Intuitionistic Truth ⇒ Intuitionistic Knowledge ⇒ Classical Truth.

They axiomatize the system IEL as follows.

1. Axioms of propositional intuitionistic logic;
2. K(A→ B)→ (KA→KB); (distribution)
3. A→KA. (co-reflection)
4. KA→¬¬A (intuitionistic reflection)

Rule Modus Ponens.

They also argue that a weaker logic of belief (IEL−) is expressed by considering
only axioms 1,2,3.

3 The G4ip prover for IPC

We will describe next our lightweight propositional intuitionistic theorem prover, that
will be used to discover an embedding of IEL into IPC.

3.1 The LJT/G4ip calculus, (restricted here to the implicational fragment)

Motivated by problems related to loop avoidance in implementing Gentzen’s LJ calcu-
lus, Roy Dyckhoff [7] introduces the following rules for the G4ip calculus3.

LJT1 : A,Γ ` A

LJT2 : A,Γ ` B
Γ ` A→B

LJT3 : B,A,Γ ` G
A→B,A,Γ ` G

3 Originally called the LJT calculus in [7]. Restricted here to its key implicational fragment.



LJT4 : D→B,Γ ` C→D B,Γ ` G
(C→D)→B,Γ ` G

Note that LJT4 ensures termination as formulas in the sequent become smaller. The
rules work with the context Γ being either a multiset or a set.

For supporting negation, one also needs to add LJT5 that deals with the special term
f alse. Then negation of A is defined as A→ f alse.

LJT5 : f alse,Γ ` G

Rules for conjunction, disjunction and bi-conditional (not shown here) are also part of
the calculus.

As it is not unusual with logic formalisms, the same calculus had been discovered
independently in the 50’s by Vorob’ev and in the 80’s-90’s by Hudelmaier [8, 9].

3.2 A Lightweight Theorem Prover for Intuitionistic Propositional Logic

Starting from the sequent calculus for the intuitionistic propositional logic in G4ip [7],
to which we have also added rules for the “<->” relation, we obtain the following
lightweight IPC prover.

:- op(525, fy, ~ ).

:- op(550, xfy, & ). % right associative

:- op(575, xfy, v ). % right associative

:- op(600, xfx, <-> ). % non associative

prove_in_ipc(T):- prove_in_ipc(T,[]).

prove_in_ipc(A,Vs):-memberchk(A,Vs),!.

prove_in_ipc(_,Vs):-memberchk(false,Vs),!.

prove_in_ipc(A<->B,Vs):-!,prove_in_ipc(B,[A|Vs]),prove_in_ipc(A,[B|Vs]).

prove_in_ipc((A->B),Vs):-!,prove_in_ipc(B,[A|Vs]).

prove_in_ipc(A & B,Vs):-!,prove_in_ipc(A,Vs),prove_in_ipc(B,Vs).

prove_in_ipc(G,Vs1):- % atomic or disj or false

select(Red,Vs1,Vs2),

prove_in_ipc_reduce(Red,G,Vs2,Vs3),

!,

prove_in_ipc(G,Vs3).

prove_in_ipc(A v B, Vs):-(prove_in_ipc(A,Vs);prove_in_ipc(B,Vs)),!.

prove_in_ipc_reduce((A->B),_,Vs1,Vs2):-!,prove_in_ipc_imp(A,B,Vs1,Vs2).

prove_in_ipc_reduce((A & B),_,Vs,[A,B|Vs]):-!.

prove_in_ipc_reduce((A<->B),_,Vs,[(A->B),(B->A)|Vs]):-!.

prove_in_ipc_reduce((A v B),G,Vs,[B|Vs]):-prove_in_ipc(G,[A|Vs]).



prove_in_ipc_imp((C->D),B,Vs,[B|Vs]):-!,prove_in_ipc((C->D),[(D->B)|Vs]).

prove_in_ipc_imp((C & D),B,Vs,[(C->(D->B))|Vs]):-!.

prove_in_ipc_imp((C v D),B,Vs,[(C->B),(D->B)|Vs]):-!.

prove_in_ipc_imp((C<->D),B,Vs,[((C->D)->((D->C)->B))|Vs]):-!.

prove_in_ipc_imp(A,B,Vs,[B|Vs]):-memberchk(A,Vs).

We validate it first by testing it on the implicational subset, derived via the Curry-
Howard isomorphism [10], then against Roy Dyckhoff’s Prolog implementation4, work-
ing on formulas up to size 12. Finally we run it on human-made tests5, on which we get
no errors, solving correctly 161 problems, with a 60 seconds timeout, compared with
the 175 problems solved by Roy Dyckhoff’s more refined, heuristics-based 400 lines
prover, with the same timeout6. We refer to [10] for the derivation steps of variants
of this prover working on the implicational and nested Horn clause fragments of IPC.
While more sophisticated tableau-based provers are available for IPC among which
we mention the excellent Prolog-based fCube [11], our prover’s compact size and ad-
equate performance will suffice, given also the space constraints imposed by literate
programming nature of this paper.

4 The definition formula generator

We start with a candidate formula generator that we will constrain further to be used for
generating candidate definitions of our modal operators.

4.1 Generating Operator Trees

We generate all formulas of a given size by decreasing the available size parameter at
each step when nodes are added to a tree representation of a formula. Prolog’s DCG
mechanism is used to collect the leaves of the tree.

genOperatorTree(N,Ops,Tree,Leaves):-

genOperatorTree(Ops,Tree,N,0,Leaves,[]).

genOperatorTree(_,V,N,N)-->[V].

genOperatorTree(Ops,OpAB,SN1,N3)-->

{ SN1>0,N1 is SN1-1,

member(Op,Ops),make_oper2(Op,A,B,OpAB)

},

genOperatorTree(Ops,A,N1,N2),

genOperatorTree(Ops,B,N2,N3).

make_oper2(Op,A,B,OpAB):-functor(OpAB,Op,2),arg(1,OpAB,A),arg(2,OpAB,B).

4 https://github.com/ptarau/TypesAndProofs/blob/master/third_party/

dyckhoff_orig.pro
5 at http://iltp.de
6 https://github.com/ptarau/TypesAndProofs/blob/master/tester.pro



4.2 Synthesizing the definitions of modal operators

As we design a generic definition discovery mechanism, we will denote generically our
modal operators as follows.

– “#” for “�”=necessary and “K”=known
– “*” for “♦”=possible and “M”=knowable

After the operator definitions

:- op( 500, fy, #).

:- op( 500, fy, *).

we specify our generator as covering the usual binary operators and we constrain it to
have at least one of the leaves of its generated trees to be a variable. Besides the false
constant used in the definition of negation, we introduce also a new constant symbol “?”
assumed not to occur in the language. Its role will be left unspecified until the possible
synthesized definitions will be filtered. We will constrain candidate definitions to ensure
that axioms and selected theorems hold and selected non-theorems fail.

genDef(M,Def):-genDef(M,[(->),(&),(v)],[false,?],Def).

genDef(M,Ops,Cs,(#(X):-T)):-

between(0,M,N),

genOperatorTree(N,Ops,T,Vs),

pick_leaves(Vs,[X|Cs]),

term_variables(Vs,[X]).

Leaves of the generated trees will be picked from a given set.

pick_leaves([],_).

pick_leaves([V|Vs],Ls):-member(V,Ls),pick_leaves(Vs,Ls).

We first expand our operator definitions for the “~” negation and “*” modal operator
while keeping atomic variables and the special constant false untouched.

expand_defs(_,false,R) :-!,R=false.

expand_defs(_,A,R) :-atomic(A),!,R= A.

expand_defs(D,~(A),(B->false)) :-!,expand_defs(D,A,B).

expand_defs(D,*(A),R):-!,expand_defs(D,~ (# (~(A))),R).

The special case for expanding a candidate operator definition D requires a fresh variable
for each instance, ensured by Prolog’s built-in copy term.

expand_defs(D,#(X),R) :-!,copy_term(D,(#(X):-T)),expand_defs(D,T,R).

Other operators are traversed generically by using Prolog’s “=..” built-in and by re-
cursing with expand def list on their arguments.

expand_defs(D,A,B) :-

A=..[F|Xs],

expand_def_list(D,Xs,Ys),

B=..[F|Ys].



expand_def_list(_,[],[]).

expand_def_list(D,[X|Xs],[Y|Ys]) :-

expand_defs(D,X,Y),

expand_def_list(D,Xs,Ys).

The predicate prove with def refines our G4ip prover by first expanding the defini-
tions extending IPC with a given candidate modality.

prove_with_def(Def,T0) :-expand_defs(Def,T0,T1),prove_in_ipc(T1,[]).

The definition synthesizer will filter the candidate definitions provided by genDef such
that the predicate prove with def succeeds on all theorems and fails on all non-
theorems, provided as names of the facts of arity 1 containing them.

def_synth(M,D):-def_synth(M,iel_th,iel_nth,D).

def_synth(M,Th,NTh,D):-

genDef(M,D),

forall(call(Th,T),prove_with_def(D,T)),

forall(call(NTh,NT), \+prove_with_def(D,NT)).

Note that the generator first builds smaller formulas and then larger ones up the specified
maximum size.

Example 1 Candidate definitions up to size 2

?- forall(genDef(2,Def),println(Def)).

#A :- A

#A :- A -> A

#A :- A -> false

#A :- A -> ?

#A :- false -> A

#A :- ? -> A

#A :- A & A

#A :- A & false

#A :- A & ?

...

#A :- (A -> ?) -> A

...

#A :- (? v A) v ?

#A :- (? v false) v A

#A :- (? v ?) v A

5 Discovering the embedding of IEL and Dosen’s double negation
modality in IPC

We specify a given logic (e.g., IEL or S4) by stating theorems on which the prover
extended with the synthetic definition should succeed and non-theorems on which it
should fail.



5.1 The discovery mechanism for IEL

We start with the 4 axioms of Artemov and Protopopescu’s IEL system:

iel_th(a -> # a).

iel_th(# (a->b)->(# a-> # b)).

iel_th(# p <-> # # p).

iel_th(# a -> ~ ~ a).

Note that the axioms would be enough to specify the logic, but we also add some theo-
rems when intuitively relevant and/or mentioned in [4].

iel_th(# (a & b) <-> (# a & # b)).

iel_th(~ # false).

iel_th(~ (# a & ~ a)).

iel_th(~a -> ~ # a).

iel_th( ~ ~ (# a -> a)).

iel_th(# a & # (a->b) -> # b).

iel_th(* (a & b) <-> (* a & * b)).

iel_th(# a -> * a).

iel_th(# a v # b -> # (a v b) ).

iel_th(* a <-> * * a).

iel_th(a -> *a).

Again, following [4], we add our non-theorems.

iel_nth(# a -> a).

iel_nth(# (a v b) -> # a v # b).

iel_nth(# a).

iel_nth(~ (# a)).

iel_nth(# false).

iel_nth(# a).

iel_nth(~ (# a)).

iel_nth(* false).

We also define (implicit) facts for supporting the necessitation rule that states that the
operator “#” applied to proven theorems or axioms generates new theorems.

iel_nec_th(T):-iel_th(T).

iel_nec_th(# T):-iel_th(T).

Finally, we obtain the discovery algorithm for IEL formula definitions and for IEL
extended with the necessitation rule.

iel_discover:-

backtrack_over((def_synth(2,iel_th,iel_nth,D),println(D))).

iel_nec_discover:-

backtrack_over((def_synth(2,iel_nec_th,iel_nth,D),println(D))).

backtrack_over(Goal):-call(Goal),fail;true.

println(T):-numbervars(T,0,_),writeln(T).



We run iel discover, ready to see the surviving definition candidates.

Example 2 Definition discovery without the necessitation rule.

?- iel_discover.

#A:-(A->false)->A

#A:-(A->false)->false

#A:-(A-> ?)->A

true.

Example 3 Definition discovery with the necessitation rule.

?- iel_nec_discover.

#A:-(A->false)->A

#A:-(A->false)->false

#A:-(A-> ?)->A

true.

Unsurprisingly, the results are the same, as a consequence of A -> #A.
Clearly, the formula #A:-(A->false)->A is not interesting as it would define

knowing something as a contradiction that implies itself.
This brings us to the second definition formula candidate.

5.2 Eliminating Dosen’s double negation modality

In [2] double negation in IPC is interpreted as a “�” modality. This corresponds to one
of the synthetic definitions #A :- (A->false)->false that is equivalent in IPC to
#A :- ~~A. It is argued in [4] that it does not make sense as an epistemic modality,
mostly because it would entail that all classical theorems are known intuitionistically.

We eliminate it by requiring the collapsing of “*” into “#” to be a non-theorem:

iel_nth(* a <-> # a).

In fact, while known (#) implies knowable (~#~ = *), it is reasonable to think, as in
most modal logics, that the inverse implication does not hold.

After that, we have:

Example 4 The double negation modality is eliminated, as it collapses # and *.

?- iel_discover.

#A:-(A -> ?)->A

true.

?- iel_nec_discover.

#A:-(A -> ?)->A

true.



5.3 Knowledge as awareness?

This leaves us with the #A :- (A -> ?) -> A.
Among the consequences of the fact that intuitionistic provability strictly implies

classical, is that there’s plenty of room left between p and ~~p, where both # and * find
their place, given that the following implication chain holds.

p -> #p -> *p -> ~~p

Let us now find an (arguably) intuitive meaning for the “?” constant in the definition.
The interpretation of knowledge as awareness about truth goes back to [12]. Our final
definition of intuitionistic epistemic modality as “#A :- (A -> ?) -> A” suggests
interpreting “?” as awareness of an agent entailed by (a proof of) A. With this in mind,
one obtains an embedding of IEL in IPC via the extension

KA ≡ (A→ eureka)→ A

where eureka is a new symbol not occurring in the language7.
In line with the Brouwer-Heyting-Kolmogorov (BHK) interpretation of intuition-

istic proof, we may say that an agent knows A iff A is validated by a proof of A that
induces awareness of the agent about it.

Thus knowledge of an agent, in this sense, collects facts that are proven construc-
tively in a way that is “understood” by the agent. The consequence

KA→¬¬A

would then simply say that intuitionistic truths, that the agent is aware of, are also
classically valid.

Thus, we can define our prover for IEL as follows.

iel_prove(P):-prove_with_def((#A :- (A -> eureka) -> A),P).

Interestingly, if one allows eureka to occur in the formulas of the language given as
input to the prover, then it becomes (the unique) value for which we have equivalence
between being known and having a proof.

?- iel_prove(#eureka <-> eureka).

true .

Similarly, it would also follow that

?- iel_prove(*eureka <-> ~ ~ eureka).

true.

Thus, one would need to forbid accepting it as part of the prover’s language to closely
follow the intended semantics of IEL.

7 Not totally accidentally named, given the way Archimedes expressed his sudden awareness
about the volume of water displaced by his immersed body.



5.4 Discussion

As the IPC fragment with two variables, implication and negation has exactly 518
equivalence classes of formulas [13, 14], one would expect the construction deriving
“*” from “#” to reach a fixpoint. We can use our prover to find out when that happens.

?- iel_prove(#p <-> ~ # (~p)).

false.

iel_prove(*p <-> ~(*(~p))).

true.

Thus the fixpoint of the construction is “*” that we have interpreted as meaning that a
proposition is knowable. Therefore, the equivalence reads reasonably that something is
knowable if and only if its negation is not knowable. Note also that

?- iel_prove(~(*(~p)) -> #p).

false.

by contrast to the equivalence �p≡ ¬♦¬p usual in classical modal logics.

6 Discovering an embedding of S4 without the necessitation rule

The fact that both IPC and S4 are known to be PSPACE-complete [15] means that
polynomial-time translations exist between them.

In fact, Gödel’s translation from IPC to S4 (by prefixing each subformula with the
� operator) shows that the embedding of IPC into S4 can be achieved quite easily, by
using purely syntactic means. However, the (very) few papers attempting the inverse
translation [16, 17] rely on methods often involving intricate semantic constructions.

We will use our definition generator to identify the problem that precludes a simple
embedding of S4 into IPC.

We start with the axioms of S4.

s4_th(# a -> a).

s4_th(# (a->b) -> (# a -> # b)).

s4_th(# a -> # # a).

We add a few theorems.

s4_th(* * a <-> * a).

s4_th(a -> * a).

s4_th(# a -> * a).

s4_th(# a v # b -> # (a v b)).

s4_th(# (a v b) -> # a v # b).

We add some non-theorems that ensure additional filtering.

s4_nth(# a).

s4_nth(~ (# a)).

s4_nth(# false).

s4_nth(* false).

s4_nth(* a -> # * a). % true only in S5

s4_nth(a -> # a).



s4_nth(* a -> a).

s4_nth(# a <-> ?).

s4_nth(* a <-> ?).

Like in the case of IEL we define implicit facts stating that the necessitation rule holds.

s4_nec_th(T):-s4_th(T).

s4_nec_th(# T):-s4_th(T).

Finally we implement the definition discovery predicates and run them.

s4_discover:-

backtrack_over((def_synth(2,s4_th,s4_nth,D),println(D))).

s4_nec_discover:-

backtrack_over((def_synth(2,s4_nec_th,s4_nth,D),println(D))).

Example 5 The necessitation rule eliminates all simple embeddings of S4 into IPC,
while a lot of definition formulas pass without it.

?- s4_discover.

#A :- A & ?

#A :- ? & A

#A :- A & (A-> ?)

#A :- A & (? -> false)

...

true.

?- s4_nec_discover.

true.

Among the definitions succeding without passing the necessity rule test, one might want
to pick #A :- ? & A as an approximation of the S4 “�” operator. In this case “?”
would simply state that “the IPC prover is sound and complete”. Still, given the failure
of the necessitation rule, the resulting logic is missing a key aspect of the intended
meaning of S4-provability.

7 Related work

Program synthesis techniques have been around in logic programming with the advent
of Inductive Logic Programming [18], but the idea of learning Prolog programs from
positive and negative examples goes back to [19]. Our definition synthesizer fits in this
paradigm, with focus on the use of a theorem prover of a decidable logic (IPC) filtering
formulas provided by a definition generator through theorems as positive examples and
non-theorems as negative examples. The idea to use the new constant “?” in our syn-
thesizer is inspired by proofs that some fragments of IPC reduced to two variables have
a (small) finite number of equivalence classes [13, 14] as well as by the introduction of
new variables, in work on polynomial embeddings of S4 into IPC [16, 17].

We refer to [4] for a thorough discussion of the merits of IEL compared to epis-
temic logics following closely classical modal logic, but the central idea about using



intuitionistic logic is that of belief and knowledge as the product of verification. Our
embedding of IEL in IPC can be seen as a simplified view of this process through a
generic “awareness of an agent” concept in line with [12].

In [1] the concept of epistemic specifications is introduced that support express-
ing knowledge and belief in an Answer Set Programming framework. Interestingly,
refinements of this work like [20] and [3] discuss difficulties related to expressing an
assumption like p→Kp in terms of ASP-based epistemic operators.

Equilibrium logic [21] gives a semantics to Answer Set programs by extending the
3-valued intermediate logic of here-and-there HT with Nelson’s constructive strong
negation. In [22] a 5-valued truth-table semantics for equilibrium logic is given. In [23]
(and several other papers) epistemic extensions of equilibrium logic [21] are proposed,
in which Kp→ p. By contrast to “alethic inspired” epistemic logics postulating Kp→ p
we closely follow the p→Kp view on which [4] is centered.

While we have eliminated Dosen’s double negation modality [5] as an epistemic
operator Kp ≡ ¬¬p, it is significant that it came out as the only other meaningful
candidate produced by our definition synthesizer.

This suggests that it might be worth investigating further how a similar definition
discovery mechanism as the one we have used for IEL and S4 would work for logics
with multiple negation operators like equilibrium logic.

Besides the Kp→ p vs. p→ Kp problem a more general question is the choice
of the logic supporting the epistemic operators, among logics with finite truth-value
models (e.g., classical logic or equilibrium logic) or, at the limit, intuitionistic logic
itself, with no such models. Arguably, this could be application dependent, as epistemic
operators built on top of IPC are likely to fit better the landscape with intricate nuances
of a richer set of epistemic and doxastic operators, while such operators built on top of
finite-valued intermediate logics would benefit from simpler decision procedures and
faster evaluation mechanisms.

8 Conclusions

We have devised a general mechanism for synthesizing definitions that extend a given
logic system endowed with a theorem prover. The set of theorems on which the ex-
tended prover should succeed and the set of non-theorems on which it should fail can
be seen as a declarative specification of the extended system. Success of the approach
on embedding the IEL system in IPC and failure on trying to embed S4 has revealed
the individual role of the axioms, theorems and rules that specify a given logic system.
Given its generality, our definition generation technique can be applied also to epistemic
or modal logic axiom systems to find out if they have interesting embeddings in ASP
and superintuitionistic logics for which high quality solvers or theorem provers exist.

Tübingen (1988)

9. Hudelmaier, J.: An O(n log n)-Space Decision Procedure for Intuitionistic Propositional
Logic. Journal of Logic and Computation 3(1) (1993) 63–75

10. Tarau, P.: A Combinatorial Testing Framework for Intuitionistic Propositional Theorem
Provers. In Alferes, J.J., Johansson, M., eds.: Practical Aspects of Declarative Languages
- 21th International Symposium, PADL 2019, Lisbon, Portugal, January 14-15, 2019, Pro-
ceedings. Volume 11372 of Lecture Notes in Computer Science., Springer (2019) 115–132

11. Ferrari, M., Fiorentini, C., Fiorino, G.: fcube: An efficient prover for intuitionistic propo-
sitional logic. In Fermüller, C.G., Voronkov, A., eds.: Logic for Programming, Artificial
Intelligence, and Reasoning, Berlin, Heidelberg, Springer Berlin Heidelberg (2010) 294–301

12. Fagin, R., Halpern, J.Y.: Belief, Awareness, and Limited Reasoning: Preliminary Report. In:
Proceedings of the 9th International Joint Conference on Artificial Intelligence - Volume 1.
IJCAI’85, San Francisco, CA, USA, Morgan Kaufmann Publishers Inc. (1985) 491–501

13. de Bruijn, N.G.: Exact finite models for minimal propositional calculus over a finite alphabet.
Technical Report 75?WSK?02, Technological University Eindhoven (November 1975)

14. Jongh, D.D., Hendriks, L., de Lavalette, G.R.R.: Computations in fragments of intuitionistic
propositional logic. J. Autom. Reasoning 7(4) (1991) 537–561

15. Statman, R.: Intuitionistic Propositional Logic is Polynomial-Space Complete. Theor. Com-
put. Sci. 9 (1979) 67–72

16. Egly, U.: A Polynomial Translation of Propositional S4 into Propositional Intuitionistic
Logic. (2007)

